
1 Overview

This document lists all the changes needed to correct problems in the TTCN implementation of test case 12.4.1.4f which is part of the UTRAN test suite. Only essential changes to the TTCN are applied and documented in section 4.

With these changes applied the test case can be demonstrated to run with one or more 3G UEs 
(see section 6). Execution log files are provided as evidence. 

3 Verification Test Summary

Test Case:
TC_12_4_1_4f
Test Group:
/12/UTRAN34_GMM.ttcn
ATS Version:
iwd-TTCN3-B2013-03_D14wk37 + essential modifications.

System Simulator used:
Rohde & Schwarz CMW500

UE used:

Verification Status:
PASS
4 Corrections required for test case 12.4.1.4f
4.1 Introduction

This section describes the changes required to make test case 12.4.1.4f run correctly with a 3G UE. All modifications are marked with label “WA#” in the TTCN comments column of the enclosed ATS [1].

The ATS version used as basis was UTRAN testsuite which is part of iwd-TTCN3-B2013-03_D14wk37 release. This testsuite provided by MCC160 contains UTRAN test cases.
4.2 Function TC_12_4_1_4f()
	Record Name
	TC_12_4_1_4f()

	Reason for change
	The current TTCN sets the guard timer to 300s. This is too small for this testcase since the main testcase itself contains a timer for 5 mins. This needs to be extended.

	Summary of change
	Increased the guard timer from 300s to 1200s.

	Source of change
	UTRAN_Testsuite.ttcn

	
	


Before:

	testcase TC_12_4_1_4f() runs on MTC_UTRAN system SYSTEM_UTRAN {
    // @purpose
    //   Routing area updating / rejected / Congestion
    var UTRAN_PTC    v_UTRAN      := null;
    var GERAN_PTC    v_GERAN      := null;
    var EUTRA_PTC    v_EUTRA      := null;
    timer t_GuardTimer := int2float(300);
    v_UTRAN := UTRAN_PTC.create alive;
    f_MTC_ConnectPTCs_UTRAN(v_UTRAN, v_GERAN, v_EUTRA);
    v_UTRAN.start(f_TC_12_4_1_4f_UTRAN());
    t_GuardTimer.start;
    f_MTC_MainLoop(t_GuardTimer);
  }


After:

	testcase TC_12_4_1_4f() runs on MTC_UTRAN system SYSTEM_UTRAN {
    // @purpose
    //   Routing area updating / rejected / Congestion
    var UTRAN_PTC    v_UTRAN      := null;
    var GERAN_PTC    v_GERAN      := null;
    var EUTRA_PTC    v_EUTRA      := null;
    timer t_GuardTimer := int2float(1200);//WA#12_4_1_4f : 300->1200
    v_UTRAN := UTRAN_PTC.create alive;
    f_MTC_ConnectPTCs_UTRAN(v_UTRAN, v_GERAN, v_EUTRA);
    v_UTRAN.start(f_TC_12_4_1_4f_UTRAN());
    t_GuardTimer.start;
    f_MTC_MainLoop(t_GuardTimer);
  }


4.3 Function fl_TC_12_4_1_4f_Body ()
	Record Name
	fl_TC_12_4_1_4f_Body ()

	Reason for change
	1) As per the expected test sequence in 34.123-1 Sec 12.4.1.4f.4 in the expected test sequence at Step 4 the UE should perform a CS registration, because of this, the TTCN implementation expects a LAU before/after ATTACH REQ. However, this is incorrect since the ATT flag is set to FALSE as per the initial conditions and hence the UE will not perform a LAU procedure. A Prose CR will be raised for this.
2) As per the expected test sequence in 34.123-1 Sec 12.4.1.4f.4 in the expected test sequence at Step 5 the SS expects an ATTACH REQ. The template used to capture the MobileIdentity in this ATTACH REQ is cs_MobileIdTMSIorPTMSI ('18'O, px_PTMSI_Def). This sets the IEI to ‘18’O. However, the IE MobileIdentity is of LVtype. The IEI should be set to omit.
3) At Step 13 and 16 of the expected test sequence, the TTCN incorrectly expects the RAI to be v_RAI_4. This needs to be corrected.
4) At Step 13, the template car_PS_UplinkDirectTransfer is used to capture the RAU REQ. This is incorrect. This should be car_PS_InitDirecTransfer since this is the first NAS message on a new rrc connection.

	Summary of change
	1) In the current TTCN, added a check for the value of ATT flag along with the OpMode , and based upon that the SS will now expect a LAU and/or ATTACH REQ. To get the status of ATT flag, declared a variable v_CellInfo, and later called the function v_CellInfo := f_UTRAN_CellInfo_Get(utran34_CellA); which will allow access to ATT flag value.
2) For the ATTACH REQ in Step 5, set the IEI for the IE MobileIdentity to OMIT. Hence, assigned v_MobileIdPTMSI_1 := cs_MobileIdTMSIorPTMSI (/*WA#12_4_1_4f '18'O*/omit, px_PTMSI_Def)
3) Instead of checking the RAI to be v_RAI_4, checked it to be v_RAI_1
4) Instead of using car_PS_UplinkDirectTransfer, used car_PS_InitDirecTransfer. Also, as an improvement, assigned an intermediate pass verdict after the Step 15.

	Source of change
	UTRAN34_GMM.ttcn

	
	


Before:

	function fl_TC_12_4_1_4f_Body () runs on UTRAN_PTC
  {
    var template (value) MobileIdentity v_MobileIdPTMSI_1;
    var RRC_DATA_IND v_RRC_DataInd_CS;
    var RRC_DATA_IND v_RRC_DataInd_PS;
    var START_Value v_START_Value_cs := '00000000000000000000'B;
    var START_Value v_START_Value_ps := '00000000000000000000'B;
    var octetstring v_PLMN_A;
    var octetstring v_PLMN_B;
    var octetstring v_LAC_A;
    var octetstring v_LAC_B;
    var octetstring v_RAC_A;
    var octetstring v_RAC_B;
    var template (present) RoutingAreaId v_RAI_1;
    var template (present) RoutingAreaId v_RAI_4;
    var UE_OperationMode v_UE_OperationMode := f_UTRAN34_MobileInfo_UE_OperationMode_Get();
    timer t_T3346:= f_UG_SetTimerToleranceMin (5.0 * 60.0); // T3346=5min    
    v_PLMN_B := f_U_Asn2Nas_PlmnId(f_UTRAN_CellInfo_GetPLMNIdentity(utran34_CellB));
    v_LAC_B := f_UTRAN_CellInfo_GetLAC(utran34_CellB);
    v_RAC_B := f_UTRAN_CellInfo_GetRAC(utran34_CellB);
    v_RAI_4 :=   cr_RAI (omit, cr_LAI(omit, v_PLMN_B, v_LAC_B), v_RAC_B);
    v_PLMN_A := f_U_Asn2Nas_PlmnId(f_UTRAN_CellInfo_GetPLMNIdentity(utran34_CellA));
    v_LAC_A := f_UTRAN_CellInfo_GetLAC(utran34_CellA);
    v_RAC_A := f_UTRAN_CellInfo_GetRAC(utran34_CellA);
    v_RAI_1 :=   cr_RAI (omit, cr_LAI(omit, v_PLMN_A, v_LAC_A), v_RAC_A);
    v_MobileIdPTMSI_1 := cs_MobileIdTMSIorPTMSI ( '18'O, px_PTMSI_Def);
    // The following messages are sent and shall be received on cell A.
    // The SS is set in network operation mode II - Done in the premable
    // Set the cell type of cell A to the "Serving cell".
    // Set the cell type of cell B to the "Non-Suitable cell".
    //@siclog "Steps 1-2" siclog@
    f_UTRAN_SetCellPower(utran34_CellA, tsc_AttenuationServingCell);
    f_UTRAN_SetCellPower(utran34_CellB, tsc_AttenuationNonSuitableCell);
    //The UE is set in UE operation mode C (see ICS)
    // done in the preamble
    //The UE is powered up or switched on and initiates an attach (see ICS). Cell A is preferred by the UE.
    //@siclog "Step 3" siclog@
    f_UT_SwitchOnUE(UT, false);
    f_UTRAN34_RRC_ConnEst (utran34_CellA);
    alt {
      [v_UE_OperationMode == opModeC] U_Dc.receive(car_PS_InitDirectTransfer(utran_CellDedicated,
                                                                             tsc_RB3,
                                                                             cr_U34_AttachReq(cr_AttachType(?, tsc_I_AttachTypeGPRSAtt),
                                                                                              v_MobileIdPTMSI_1,
                                                                                              v_RAI_1,
                                                                                              ?))) -> value v_RRC_DataInd_PS
        {
          // ATTACH REQUEST
          // Attach type = 'GPRS attach'-Mobile identity =P-TMSI-1-Old Routing area identity = RAI-1
          //@siclog "Step 5" siclog@
          v_START_Value_ps := v_RRC_DataInd_PS.ttcn_start;
          f_UTRAN_SS_SecurityDownloadStart(ps_domain, v_START_Value_ps);
        }
        [v_UE_OperationMode == opModeA ] U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U_LocationUpdatingRequest))
            -> value v_RRC_DataInd_CS
          {
            // Registration on CS
            // See TS 34.108 This step is applied only for UE in UE operation mode A.
            //@siclog "Step 4" siclog@
            v_START_Value_cs := v_RRC_DataInd_CS.ttcn_start;
            f_UTRAN_SS_SecurityDownloadStart(cs_domain, v_START_Value_cs);
            if (pc_AutomaticAttachSwitchON) {
              // Auto attach UE at switch on
              // Need to wait for Attach Req before finishing LAU
              // ATTACH REQUEST
              // Attach type = 'GPRS attach'-Mobile identity =P-TMSI-1-Old Routing area identity = RAI-1
              //@siclog "Step 5" siclog@
              U_Dc.receive(car_PS_InitDirectTransfer(utran_CellDedicated,
                                                     tsc_RB3,
                                                     cr_U34_AttachReq(cr_AttachType(?, tsc_I_AttachTypeGPRSAtt),
                                                                      v_MobileIdPTMSI_1,
                                                                      v_RAI_1,
                                                                      ?))) -> value v_RRC_DataInd_PS;
              v_START_Value_ps := v_RRC_DataInd_PS.ttcn_start;
              f_UTRAN_SS_SecurityDownloadStart(ps_domain, v_START_Value_ps);
              f_UTRAN_LocationUpdate_WithoutLAUReq(utran34_CellA, v_START_Value_cs, true);
            } else {
              // Non auto attach UE at switch on
              // Finish LAU first
              f_UTRAN_LocationUpdate_WithoutLAUReq(utran34_CellA, v_START_Value_cs, true);
              // then trigger the Attach Req
              f_UTRAN_RRC_ConnRel(utran34_CellA, cell_Dch);
              f_Delay(1.0);
              f_UT_TriggerAttach(UT);
              f_UTRAN_RRC_ConnEst(utran34_CellA); // Establish RRC connection
              // Attach type = 'GPRS attach'-Mobile identity =P-TMSI-1-Old Routing area identity = RAI-1
              //@siclog "Step 5" siclog@
              U_Dc.receive(car_PS_InitDirectTransfer(utran_CellDedicated,
                                                     tsc_RB3,
                                                     cr_U34_AttachReq(cr_AttachType(?, tsc_I_AttachTypeGPRSAtt),
                                                                      v_MobileIdPTMSI_1,
                                                                      v_RAI_1,
                                                                      ?))) -> value v_RRC_DataInd_PS;
              v_START_Value_ps := v_RRC_DataInd_PS.ttcn_start;
              f_UTRAN_SS_SecurityDownloadStart(ps_domain, v_START_Value_ps);
            }
          }
    }
    // AUTHENTICATION AND CIPHERING REQUEST - AUTHENTICATION AND CIPHERING RESPONSE
    //@siclog "Steps 6-7" siclog@
    f_UTRAN_GMM_Authentication();
    // The SS starts integrity protection.
    //@siclog "Step 8" siclog@
    f_UTRAN34_RRC_Security(utran34_CellA,     // SECURITY MODE COMMAND, SECURITY MODE COMPLETE
                           true,  //New key
                           ps_domain);
    //@siclog "Steps 9-10" siclog@
    // ATTACH ACCEPT: Attach result = 'GPRS only attached' Allocated P-TMSI = P-TMSI-2 P-TMSI Signature = P-TMSI-2 signature Routing area identity = RAI-1
    f_UTRAN34_AttachAccept(utran34_CellA, v_RRC_DataInd_PS.msg.attachRequest.attachType.attachType, tsc_PTMSI_2, tsc_PTMSISig_2); // ATTACH ACCEPT, ATTACH COMPLETE
    f_UTRAN34_RRC_ConnRel(utran34_CellA, cell_Dch);
    // Set the cell type of cell A to the " Suitable neighbour cell".
    // Set the cell type of cell B to the "Serving cell".
    // Cell B is preferred by the UE.
    //@siclog "Steps 11" siclog@
    f_UTRAN_SetCellPower(utran34_CellA, tsc_AttenuationSuitableNeighbourCell);
    f_UTRAN_SetCellPower(utran34_CellB, tsc_AttenuationServingCell);
    //Configure cell for RRC Connection establishment
    f_UTRAN_PrepareCellRRC_ConnEst(utran34_CellB, utran34_CellA);
    f_UTRAN34_RRC_ConnEst (utran34_CellB);
    //@siclog "Step 13" siclog@
    /* ROUTING AREA UPDATE REQUEST - Update type = 'RA updating' - Old P-TMSI signature=P-TMSI-2 signature - Old Routing area identity = RAI-1  */
    U_Dc.receive(car_PS_UplinkDirectTransfer(utran_CellDedicated,
                                             tsc_RB3,
                                             cr_U_RA_UpdReqAny(cr_GMM_UpdateType(?, tsc_I_UpdateTypeRAOnly),
                                                               v_RAI_4,
                                                               cr_PTMSI_Signature(tsc_PTMSISig_2) ,
                                                               ?))) -> value v_RRC_DataInd_PS;
    v_START_Value_ps := v_RRC_DataInd_PS.ttcn_start;
    f_UTRAN_SS_SecurityDownloadStart(ps_domain, v_START_Value_ps);
    //@siclog "Step 13A" siclog@
    // The SS starts integrity protection
    f_UTRAN34_RRC_Security(utran34_CellB,     // SECURITY MODE COMMAND, SECURITY MODE COMPLETE
                           false,  //No New key
                           ps_domain);
    // ROUTING AREA UPDATE REJECT - GMM cause = Congestion - T3346 value = 5 minutes */
    //@siclog "Step 14" siclog@
    U_Dc.send(cas_DataReq_PS(utran_CellDedicated,
                             tsc_RB3,
                             cs_RAUpdReject_T3346)); // cause Congestion
    t_T3346.start;
    f_UTRAN34_RRC_ConnRel(utran34_CellB, cell_Dch);
    // The SS verifies that the MS does not initiate the routing area update procedure before timer T3346 has expired
    //@siclog "Step 14" siclog@    
    alt {
      [] U_TM.receive(car_RRC_ConnReq(?,
                                      tsc_RB0,
                                      cr_108_RRC_ConnReq_r8(?)))
        {
          f_UTRAN_SetVerdictFailOrInconc (__FILE__, __LINE__, "Step 14");
        }
      [] t_T3346.timeout
        {}
    }    
    // ROUTING AREA UPDATE REQUEST
    // Update type = RA updating - Old P-TMSI signature=P-TMSI-2 signature Old Routing area identity = RAI-1
    //@siclog "Steps 15-18" siclog@
    f_UTRAN34_GMM_RAU( utran34_CellB,
                       registration,
                       -,
                       v_RAI_4,

                       px_PTMSI_Def,
                       tsc_PTMSISig_2,
                       -,
                       false //No authentication
                       )
  }  // fl_TC_12_4_1_4f_Body
}
	


After:

	function fl_TC_12_4_1_4f_Body () runs on UTRAN_PTC
  {
    var template (value) MobileIdentity v_MobileIdPTMSI_1;
    var RRC_DATA_IND v_RRC_DataInd_CS;
    var RRC_DATA_IND v_RRC_DataInd_PS;
    var START_Value v_START_Value_cs := '00000000000000000000'B;
    var START_Value v_START_Value_ps := '00000000000000000000'B;
    var octetstring v_PLMN_A;
    var octetstring v_PLMN_B;
    var octetstring v_LAC_A;
    var octetstring v_LAC_B;
    var octetstring v_RAC_A;
    var octetstring v_RAC_B;
    var template (present) RoutingAreaId v_RAI_1;
    var template (present) RoutingAreaId v_RAI_4;
    var UE_OperationMode v_UE_OperationMode := f_UTRAN34_MobileInfo_UE_OperationMode_Get();
    var UTRAN_CellInfo_Type v_CellInfo; //WA#12_4_1_4f
    timer t_T3346:= f_UG_SetTimerToleranceMin (5.0 * 60.0); // T3346=5min    
    v_PLMN_B := f_U_Asn2Nas_PlmnId(f_UTRAN_CellInfo_GetPLMNIdentity(utran34_CellB));
    v_LAC_B := f_UTRAN_CellInfo_GetLAC(utran34_CellB);
    v_RAC_B := f_UTRAN_CellInfo_GetRAC(utran34_CellB);
    v_RAI_4 :=   cr_RAI (omit, cr_LAI(omit, v_PLMN_B, v_LAC_B), v_RAC_B);
    v_PLMN_A := f_U_Asn2Nas_PlmnId(f_UTRAN_CellInfo_GetPLMNIdentity(utran34_CellA));
    v_LAC_A := f_UTRAN_CellInfo_GetLAC(utran34_CellA);
    v_RAC_A := f_UTRAN_CellInfo_GetRAC(utran34_CellA);
    v_RAI_1 :=   cr_RAI (omit, cr_LAI(omit, v_PLMN_A, v_LAC_A), v_RAC_A);
    v_MobileIdPTMSI_1 := cs_MobileIdTMSIorPTMSI (/*WA#12_4_1_4f '18'O*/omit, px_PTMSI_Def);
    // The following messages are sent and shall be received on cell A.
    // The SS is set in network operation mode II - Done in the premable
    // Set the cell type of cell A to the "Serving cell".
    // Set the cell type of cell B to the "Non-Suitable cell".
    //@siclog "Steps 1-2" siclog@
    f_UTRAN_SetCellPower(utran34_CellA, tsc_AttenuationServingCell);
    f_UTRAN_SetCellPower(utran34_CellB, tsc_AttenuationNonSuitableCell);
    v_CellInfo := f_UTRAN_CellInfo_Get(utran34_CellA);//WA#12_4_1_4f
    //The UE is set in UE operation mode C (see ICS)
    // done in the preamble
    //The UE is powered up or switched on and initiates an attach (see ICS). Cell A is preferred by the UE.
    //@siclog "Step 3" siclog@
    f_UT_SwitchOnUE(UT, false);
    f_UTRAN34_RRC_ConnEst (utran34_CellA);
    alt {
      [v_UE_OperationMode == opModeC] U_Dc.receive(car_PS_InitDirectTransfer(utran_CellDedicated,
                                                                             tsc_RB3,
                                                                             cr_U34_AttachReq(cr_AttachType(?, tsc_I_AttachTypeGPRSAtt),
                                                                                              v_MobileIdPTMSI_1,
                                                                                              v_RAI_1,
                                                                                              ?))) -> value v_RRC_DataInd_PS
        {
          // ATTACH REQUEST
          // Attach type = 'GPRS attach'-Mobile identity =P-TMSI-1-Old Routing area identity = RAI-1
          //@siclog "Step 5" siclog@
          v_START_Value_ps := v_RRC_DataInd_PS.ttcn_start;
          f_UTRAN_SS_SecurityDownloadStart(ps_domain, v_START_Value_ps);
        }
        [v_UE_OperationMode == opModeA and /*WA#12_4_1_4a*/v_CellInfo.attFlag == tsc_AttOn ] U_Dc.receive(car_InitDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U_LocationUpdatingRequest))
            -> value v_RRC_DataInd_CS
          {
            // Registration on CS
            // See TS 34.108 This step is applied only for UE in UE operation mode A.
            //@siclog "Step 4" siclog@
            v_START_Value_cs := v_RRC_DataInd_CS.ttcn_start;
            f_UTRAN_SS_SecurityDownloadStart(cs_domain, v_START_Value_cs);
            if (pc_AutomaticAttachSwitchON) {
              // Auto attach UE at switch on
              // Need to wait for Attach Req before finishing LAU
              // ATTACH REQUEST
              // Attach type = 'GPRS attach'-Mobile identity =P-TMSI-1-Old Routing area identity = RAI-1
              //@siclog "Step 5" siclog@
              U_Dc.receive(car_PS_InitDirectTransfer(utran_CellDedicated,
                                                     tsc_RB3,
                                                     cr_U34_AttachReq(cr_AttachType(?, tsc_I_AttachTypeGPRSAtt),
                                                                      v_MobileIdPTMSI_1,
                                                                      v_RAI_1,
                                                                      ?))) -> value v_RRC_DataInd_PS;
              v_START_Value_ps := v_RRC_DataInd_PS.ttcn_start;
              f_UTRAN_SS_SecurityDownloadStart(ps_domain, v_START_Value_ps);
              f_UTRAN_LocationUpdate_WithoutLAUReq(utran34_CellA, v_START_Value_cs, true);
            } else {
              // Non auto attach UE at switch on
              // Finish LAU first
              f_UTRAN_LocationUpdate_WithoutLAUReq(utran34_CellA, v_START_Value_cs, true);
              // then trigger the Attach Req
              f_UTRAN_RRC_ConnRel(utran34_CellA, cell_Dch);
              f_Delay(1.0);
              f_UT_TriggerAttach(UT);
              f_UTRAN_RRC_ConnEst(utran34_CellA); // Establish RRC connection
              // Attach type = 'GPRS attach'-Mobile identity =P-TMSI-1-Old Routing area identity = RAI-1
              //@siclog "Step 5" siclog@
              U_Dc.receive(car_PS_InitDirectTransfer(utran_CellDedicated,
                                                     tsc_RB3,
                                                     cr_U34_AttachReq(cr_AttachType(?, tsc_I_AttachTypeGPRSAtt),
                                                                      v_MobileIdPTMSI_1,
                                                                      v_RAI_1,
                                                                      ?))) -> value v_RRC_DataInd_PS;
              v_START_Value_ps := v_RRC_DataInd_PS.ttcn_start;
              f_UTRAN_SS_SecurityDownloadStart(ps_domain, v_START_Value_ps);
            }
          }
          //WA#12_4_1_4f : Start
          [v_UE_OperationMode == opModeA and v_CellInfo.attFlag == tsc_AttOff ] U_Dc.receive(car_PS_InitDirectTransfer(utran_CellDedicated,
                                                     tsc_RB3,
                                                     cr_U34_AttachReq (cr_AttachType(?, tsc_I_AttachTypeGPRSAtt),
                                                                      v_MobileIdPTMSI_1,
                                                                      v_RAI_1,
                                                                      ?))) -> value v_RRC_DataInd_PS
          {
            v_START_Value_ps := v_RRC_DataInd_PS.ttcn_start;
            f_UTRAN_SS_SecurityDownloadStart(ps_domain, v_START_Value_ps);            
          }
    }
    // AUTHENTICATION AND CIPHERING REQUEST - AUTHENTICATION AND CIPHERING RESPONSE
    //@siclog "Steps 6-7" siclog@
    f_UTRAN_GMM_Authentication();
    // The SS starts integrity protection.
    //@siclog "Step 8" siclog@
    f_UTRAN34_RRC_Security(utran34_CellA,     // SECURITY MODE COMMAND, SECURITY MODE COMPLETE
                           true,  //New key
                           ps_domain);
    //@siclog "Steps 9-10" siclog@
    // ATTACH ACCEPT: Attach result = 'GPRS only attached' Allocated P-TMSI = P-TMSI-2 P-TMSI Signature = P-TMSI-2 signature Routing area identity = RAI-1
    f_UTRAN34_AttachAccept(utran34_CellA, v_RRC_DataInd_PS.msg.attachRequest.attachType.attachType, tsc_PTMSI_2, tsc_PTMSISig_2); // ATTACH ACCEPT, ATTACH COMPLETE
    f_UTRAN34_RRC_ConnRel(utran34_CellA, cell_Dch);
    // Set the cell type of cell A to the " Suitable neighbour cell".
    // Set the cell type of cell B to the "Serving cell".
    // Cell B is preferred by the UE.
    //@siclog "Steps 11" siclog@
    f_UTRAN_SetCellPower(utran34_CellA, tsc_AttenuationSuitableNeighbourCell);
    f_UTRAN_SetCellPower(utran34_CellB, tsc_AttenuationServingCell);
    //Configure cell for RRC Connection establishment
    f_UTRAN_PrepareCellRRC_ConnEst(utran34_CellB, utran34_CellA);
    f_UTRAN34_RRC_ConnEst (utran34_CellB);
    //@siclog "Step 13" siclog@
    /* ROUTING AREA UPDATE REQUEST - Update type = 'RA updating' - Old P-TMSI signature=P-TMSI-2 signature - Old Routing area identity = RAI-1  */
    U_Dc.receive(car_PS_InitDirectTransfer/*WA#12_4_1_4f car_PS_UplinkDirectTransfer*/(utran_CellDedicated,
                                             tsc_RB3,
                                             cr_U_RA_UpdReqAny(cr_GMM_UpdateType(?, tsc_I_UpdateTypeRAOnly),
                                                               v_RAI_1/*WA#12_4_1_4f v_RAI_4*/,
                                                               cr_PTMSI_Signature(tsc_PTMSISig_2) ,
                                                               ?))) -> value v_RRC_DataInd_PS;
    v_START_Value_ps := v_RRC_DataInd_PS.ttcn_start;
    f_UTRAN_SS_SecurityDownloadStart(ps_domain, v_START_Value_ps);
    //@siclog "Step 13A" siclog@
    // The SS starts integrity protection
    f_UTRAN34_RRC_Security(utran34_CellB,     // SECURITY MODE COMMAND, SECURITY MODE COMPLETE
                           false,  //No New key
                           ps_domain);
    // ROUTING AREA UPDATE REJECT - GMM cause = Congestion - T3346 value = 5 minutes */
    //@siclog "Step 14" siclog@
    U_Dc.send(cas_DataReq_PS(utran_CellDedicated,
                             tsc_RB3,
                             cs_RAUpdReject_T3346)); // cause Congestion
    t_T3346.start;
    f_UTRAN34_RRC_ConnRel(utran34_CellB, cell_Dch);
    // The SS verifies that the MS does not initiate the routing area update procedure before timer T3346 has expired
    //@siclog "Step 14" siclog@    
    alt {
      [] U_TM.receive(car_RRC_ConnReq(?,
                                      tsc_RB0,
                                      cr_108_RRC_ConnReq_r8(?)))
        {
          f_UTRAN_SetVerdictFailOrInconc (__FILE__, __LINE__, "Step 14");
        }
      [] t_T3346.timeout
        {
          f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 15");//WA#12_4_1_4f
        }
    }    
    // ROUTING AREA UPDATE REQUEST
    // Update type = RA updating - Old P-TMSI signature=P-TMSI-2 signature Old Routing area identity = RAI-1
    //@siclog "Steps 15-18" siclog@
    f_UTRAN34_GMM_RAU( utran34_CellB,
                       registration,
                       -,
                       v_RAI_1,//WA#12_4_1_4f v_RAI_4
                       px_PTMSI_Def,
                       tsc_PTMSISig_2,
                       -,
                       false //No authentication
                       )
  }  // fl_TC_12_4_1_4f_Body
}
	


5 Branches executed in test case 12.4.1.4f
The test case implementation was executed with Integrity and ciphering enabled.

6 Execution Log Files

6.1 Qualcomm 8974 Pro UE
The Qualcomm 8974 Pro UE passed this test case on Rohde & Schwarz CMW500. The documentation below is enclosed as evidence of the successful test case run [1]:

Test Case Execution log file: 
Rohde-Schwarz\tc_12_4_1_4f_Qualcomm_8974_Pro.log 
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)
In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file

